Re_rate_err

In the wholesale long distance business, routes and rates are constantly changing. We maintain a near real-time view of our routes’ profitability by rating every call almost as soon as it’s completed. Occassionally, a new route will come online in the switch before we’ve been notified of it. This will send calls to an exception table which need to be revisited when we have the new route information.

The following routine is part of a larger package correcting or updating rated information. This was first a nightly, then an hourly routine. When bulk operations were introduced, this was one of the first routines I rewrote.

The old version used a temporary table to hold a list of the CDR ID’s which had rerated successfully. That table would be inserted, queried and deleted as we processed blocks of calls in the exception table. Some calls would still not be able to rate and have to go back to the exception table. There was an exceptional amount of DML going on.

The new version uses collections as its temporary holding areas for successful and failed CDR. The time to completion for each run improved by 60%.

PROCEDURE re_rate_err
(  ior_hist_rater IN OUT hist_rater%ROWTYPE
   ,on_rc       OUT t.t_return_cd
)
IS
   /* cursor snipped */

   ln_tx_idx            PLS_INTEGER := 0;
   ln_bgn_cdr_id        cdr.cdr_id%TYPE;
   lr_cdr_info          rate.tr_cdr_info;
   lr_cust_totals       rate.tr_cust_totals;
   lr_vend_totals       rate.tr_vend_totals;
   ln_rc                t.t_return_cd;
   lr_cdr_rated         cdr_rated%ROWTYPE;
   
   -- local array types, to be used by the stacks   
   TYPE tan_cdr IS TABLE OF cdr_rated.cdr_id%TYPE
      INDEX BY BINARY_INTEGER;
   TYPE tan_err_id IS TABLE OF x_cdr_rate_err.error_id%TYPE
      INDEX BY BINARY_INTEGER;

   -- error stack, used for bulk updating, index subscript must be kept in sync
   -- to work for bulk binding
   lan_err_cdr tan_cdr; -- holds cdr IDs to be updated
   lan_err_id  tan_err_id; -- holds error IDs to attach to cdr IDs

   -- success stack, used for bulk deleting from rate err table
   lan_cdr tan_cdr; -- holds cdr IDs to be deleted

   -- local exceptions used for breaking processing inside loops (prevents lots
   -- of nested IF statements
   lx_rate_err       EXCEPTION;

   PROCEDURE do_savepoint
   IS
   BEGIN
      -- delete successfully rated CDR from xcre
      IF (lan_cdr.COUNT > 0) THEN
         FORALL i IN lan_cdr.FIRST..lan_cdr.LAST
            DELETE FROM x_cdr_rate_err
            WHERE cdr_id = lan_cdr(i);
         COMMIT;
      END IF;

      -- update cdr's already in xcre with new error IDs
      IF (lan_err_cdr.COUNT > 0) THEN
         FORALL i IN lan_err_cdr.FIRST..lan_err_cdr.LAST
            UPDATE x_cdr_rate_err
            SET error_id = lan_err_id(i)
            WHERE cdr_id = lan_err_cdr(i);
         COMMIT;
      END IF;
      -- empty the stacks
      lan_err_cdr.DELETE;
      lan_err_id.DELETE;
      lan_cdr.DELETE;

      -- Note, since we use the NVL(collection.LAST,0)+1 method of determining
      -- each collection's next subscript, there are no separate counters to
      -- reset back to zero.
   END do_savepoint;

BEGIN
   /* control and logging code snipped */

   -- re-initialize the counter
   ln_tx_idx := 0;

   -- initialize the return code
   ln_rc := c.SUCCESS;

   -- select from cdr the info needed to process each rateable call
   FOR lr_cdr IN cur_cdr LOOP

      ln_tx_idx := ln_tx_idx + 1;

      BEGIN
         -- if we don't have assoc ID's, don't bother to rate
         IF (lr_cdr.assoc_id_in IS NULL OR lr_cdr.assoc_id_out IS NULL) THEN
            ior_hist_rater.num_ztg_rate_err :=
               ior_hist_rater.num_ztg_rate_err + 1;
            ln_rc := rate.NO_MATCH_ZTG;
            RAISE lx_rate_err;
         END IF;

         /* sensitive code snipped */

         -- load up the record required by the get totals procedures
         lr_cdr_info.dialed_num := lr_cdr.clean_dialed_num;
         lr_cdr_info.bgn_dtm := lr_cdr.bgn_dtm;
         lr_cdr_info.assoc_id_in  := lr_cdr.assoc_id_in;
         lr_cdr_info.assoc_id_out := lr_cdr.assoc_id_out;
         lr_cdr_info.call_secs := lr_cdr.call_secs;
         lr_cdr_info.switch_id := lr_cdr.switch_id;
         lr_cdr_info.re_rate := 'N';

         rate.get_rvn_totals(lr_cdr_info, lr_cust_totals, ln_rc);

         IF (ln_rc <> c.SUCCESS) THEN
            IF (ln_rc = rate.NO_MATCH_CRM) THEN
               ior_hist_rater.num_crm_rate_err := 
                  ior_hist_rater.num_crm_rate_err + 1;
            ELSIF (ln_rc = rate.NO_MATCH_CR) THEN
               ior_hist_rater.num_cr_rate_err := 
                  ior_hist_rater.num_cr_rate_err + 1;
            ELSE
               NULL; -- currently no call to do anything differently
            END IF;
            RAISE lx_rate_err;
         END IF;

         rate.get_cost_totals(lr_cdr_info, lr_vend_totals, ln_rc);

         IF (ln_rc <> c.SUCCESS) THEN
            IF (ln_rc = rate.NO_MATCH_VCM) THEN
               ior_hist_rater.num_vcm_rate_err := 
                  ior_hist_rater.num_vcm_rate_err + 1;
            ELSIF (ln_rc = rate.NO_MATCH_VC) THEN
               ior_hist_rater.num_vc_rate_err := 
                  ior_hist_rater.num_vc_rate_err + 1;
            ELSE
               NULL; -- currently no call to do anything differently
            END IF;
            RAISE lx_rate_err;
         END IF;


         -- write to cdr_rated
         lr_cdr_rated.cdr_id := lr_cdr.cdr_id;
         lr_cdr_rated.combo_cd := lr_cdr.combo_cd;
         lr_cdr_rated.trg_cd_in := lr_cdr.trg_cd_in;
         lr_cdr_rated.trg_cd_out := lr_cdr.trg_cd_out;
         lr_cdr_rated.rate_id := lr_cust_totals.rate_id;
         lr_cdr_rated.cost_id := lr_vend_totals.cost_id;
         lr_cdr_rated.call_dth := lr_cdr.call_dth;
         lr_cdr_rated.bgn_dtm := lr_cdr.bgn_dtm;
         lr_cdr_rated.dialed_num := lr_cdr.clean_dialed_num;
         lr_cdr_rated.call_secs := lr_cdr.call_secs;
         lr_cdr_rated.rvn_min := lr_cust_totals.rvn_min;
         lr_cdr_rated.total_rvn := lr_cust_totals.total_rvn;
         lr_cdr_rated.cost_min := lr_vend_totals.cost_min;
         lr_cdr_rated.total_cost := lr_vend_totals.total_cost;
         lr_cdr_rated.cdr_id_out := NULL;
         lr_cdr_rated.rater_id := ior_hist_rater.rater_id;

         -- first attempt to insert (more common that it's not in cdr_rated)
         api_cdr_rated.ins(lr_cdr_rated, ln_rc);
         IF (ln_rc <> c.SUCCESS) THEN
            api_cdr_rated.upd(lr_cdr_rated, ln_rc);
         END IF;
         ior_hist_rater.num_cdr_rated := ior_hist_rater.num_cdr_rated +
               (utils.fn_ite((ln_rc = c.SUCCESS),1,0));
         -- We assume the insert worked and don't check the rc.
         -- Any insertion errors will not be caught and bubble up, crashing the
         -- module so that we can debug it before it causes further harm.

         -- add the ID to the sucess stack for later deletion
         lan_cdr(NVL(lan_cdr.LAST,0)+1) := lr_cdr.cdr_id;

      EXCEPTION
         WHEN lx_rate_err THEN
            lan_err_cdr(NVL(lan_err_cdr.LAST,0)+1) := lr_cdr.cdr_id;
            lan_err_id(NVL(lan_err_id.LAST,0)+1) := ln_rc;

      END; -- end of the body for the current CDR

      IF (MOD(ln_tx_idx, 10000) = 0) THEN
         /* more logging code snipped */
         
         -- empty error and success stacks, removing/updating cdr's in xcre
         do_savepoint;
      END IF;

   END LOOP;

   -- empty error and success stacks, removing/updating remaining cdr's in xcre
   do_savepoint;

   /* more logging code snipped */

   -- load up audit rating record with end process metadata
   ior_hist_rater.num_cdr := ln_tx_idx;
   ior_hist_rater.end_dtm := SYSDATE;
   on_rc := ln_rc;

END re_rate_err;


get_groups

Using the new dbms_ldap package requires a solid understanding of associative arrays. In particular, the explode_dn routine returns an array of string that oddly places its first entry at subscript 0. I’d forgotten that you could do zero and negative subscripts with associative arrays. It took me a good hour to figure out what was going on and replace the original subscript with the .FIRST method. 

PROCEDURE get_groups
(
   is_dn     IN    VARCHAR2 -- should be DN of group OU
   ,oas_groups OUT   t.tas80 -- pl/sql table of varchar2(80)
)
IS
   l_session    dbms_ldap.SESSION; -- handle to LDAP server
   l_results    dbms_ldap.MESSAGE; --handle to LDAPMessage envelope
   las_vals     dbms_ldap.STRING_COLLECTION; --for breaking apart the DN
   
   -- vars for handling entry iteration
   l_entry      dbms_ldap.MESSAGE;
   ln_entry_idx PLS_INTEGER := 0;
   las_attrs    dbms_ldap.STRING_COLLECTION;
   
   ln_rc        PLS_INTEGER := 0;
   lx_session_failure       EXCEPTION;
   
BEGIN
   -- init session and bind anonymously to dir server
   start_ldap_sess(l_session); -- session var is IN/OUT
   
   las_attrs(1) := 'dn'; -- list of attributes desired
   
   -- execute our search to find all group entries under a given DN.
   handle_rc(
      dbms_ldap.search_s(
         l_session -- handle to LDAP server comm session
         ,is_dn -- starting point of search in DIT
         ,dbms_ldap.SCOPE_SUBTREE -- how deep to search
         ,'(objectclass=groupOfUniqueNames)' -- filter
         ,las_attrs -- array of attributes desired in the result
         ,0 -- 0=attributes and their values; non-zero=attributes only)
         ,l_results
      ), 'search');

   IF (dbms_ldap.count_entries(l_session, l_results) > 0) THEN -- iterate through any entries
      
      -- MUST use first_entry func to get the required arg for later call to next_entry
      l_entry := dbms_ldap.first_entry(l_session, l_results);
      
      WHILE (l_entry IS NOT NULL) LOOP      
         ln_entry_idx := ln_entry_idx + 1;   
         -- get DN from entry, and tokenize the parts...
         las_vals := dbms_ldap.explode_dn(
            dbms_ldap.get_dn(l_session, l_entry)
            ,1 -- "0" leaves attribute in token, e.g. 'ou=', "1" skims off the attributes types
         );
         -- ...to place group name from first RDN into outbound result
         oas_groups(ln_entry_idx) := las_vals(las_vals.FIRST);
      
         l_entry := dbms_ldap.next_entry(l_session, l_entry);
      END LOOP;
   ELSE
      NULL; -- leave OUT array empty
   END IF; -- if entries returned from search
      
   stop_ldap_sess(l_session); -- private pkg func that unbinds

END get_groups;

Api_objlock.del

This is part of a larger package, the back end to a 9iAS/PLSQL-based web screen built around a pessimistic locking table. The web screen allows an administrator to remove locks on certain editable items in our various systems (happens when a technician uses the browser’s X to close the window, rather than the provided Cancel/Close buttons which we can use in the submit routine to release the lock). The administrator checks the checkbox next to any locked item and submits the Delete request. After a layer or two of intervening code, the list of requested item Ids ends up here, being passed in as an array of number.

This routine is an example of a prime candidate for replacement with BULK delete.

/* Old */
PROCEDURE del
(
   ian_lock_id   IN t.number_table,
   on_rc    OUT   t.t_return_cd
)
IS
   ln_rc t.t_return_cd;
BEGIN
   FOR i IN 1..ian_lock_id.COUNT LOOP
      del(ian_lock_id(i), ln_rc);
      IF (ln_rc <> c.SUCCESS) THEN
         ROLLBACK;
         EXIT;
      END IF;
   END LOOP;
   on_rc := ln_rc;
END del;


/* New */
PROCEDURE del
(
   ian_lock_id   IN t.number_table
)
IS
BEGIN
   FORALL i IN ian_lock_id.FIRST..ian_lock_id.LAST
      DELETE FROM objlock
      WHERE objlock.lock_id = ian_lock_id(i);
END del;


tt.tt_search_rslt

Another routine that makes heavy use of associative arrays. This is the routine called directly from a 9iAS/PLSQL-based web form meant for composing a dynamic report.

PROCEDURE tt_search_rslt
(
    ias_col_nm  IN t.vc30_table DEFAULT t.empty_vc30
   ,ias_sort_val   IN t.vc30_table DEFAULT t.empty_vc30
   ,ias_sort_direction   IN t.vc5_table DEFAULT t.empty_vc5
   ,ias_query_item_nm IN t.vc30_table DEFAULT t.empty_vc30
   ,ias_query_op  IN t.vc30_table DEFAULT t.empty_vc30
   ,ias_query_val IN t.vc2000_table DEFAULT t.empty_vc2000
)
IS
   lar_tt_rpt  tt_d.tar_tt_rpt;
   ld_date_buf DATE;
   lar_inp_errs   ia_inp_val.tar_input_err;
BEGIN
   -- STANDARD SECURITY GATE
   IF NOT (ia_sess.fb_authorized) THEN
      RETURN;
   END IF;
   
   -- INPUT VALIDATION SECTION 
   ld_date_buf := ia_inp_val.fd_check_date(ias_query_val(12),'Date Opened',
      lar_inp_errs);
   ld_date_buf := ia_inp_val.fd_check_date(ias_query_val(13),'Date Closed',
      lar_inp_errs);
   IF (ia_inp_val.fb_check_errors(lar_inp_errs)) THEN
      RETURN; -- sends error HTML page to user and does not continue
   END IF;
   
   -- PROCESS REQUEST FOR DATA
   -- data layer will use dynamic SQL to construct the statement, query and
   -- place the result in the table of records.
   tt_d.get_tt_search_rslt
   (
      ias_sort_val,
      ias_sort_direction,
      ias_query_item_nm,
      ias_query_op,
      ias_query_val,
      lar_tt_rpt -- OUT variable, a PL/SQL table of records
   );

   -- initializing the map sets up a PL/SQL "hash table" mapping column names
   -- to their visual counterpart (a report column header label) and their
   -- relative positions to one another
   tt_d.init_map;
   
   -- display result set (uses ias_col_nm to know which columns to include)
   ia_ui.set_font_size(1);
   ia_ui.header('Ad-Hoc TT Reports: Search Results',FALSE);
   ia_ui.report_header('Trouble Ticket Report', TRUE, TRUE);   
   ia_ui.tableOpen;
      
      htp.tableRowOpen;
         -- display the headers only for the columns the user wanted to see
         FOR i IN 1..ias_col_nm.COUNT LOOP
            ia_ui.tcell_header(is_str => ia_ui.s(tt_d.fs_get_hdr(ias_col_nm(i)),
               '2'), is_wrap => 'Y');
         END LOOP;
      htp.tableRowClose;
   
   
      -- open cursor and loop through each item in the result set
      FOR j IN 1..lar_tt_rpt.COUNT LOOP  
         htp.tableRowOpen;
         
         -- again read columns the user picked in order to fetch only the
         -- desired items from the result set
         FOR i IN 1..ias_col_nm.COUNT LOOP
         
            IF (i = 1 AND ias_col_nm(i) = 'tick_id') THEN
               ia_ui.tcell_str
                (
                  htf.anchor2
                  (
                     ia_ui.fs_encode_query('tt.mod_tt?is_tick_id='||
                        TO_CHAR(lar_tt_rpt(j).tick_id)),
                     ia_ui.s(lar_tt_rpt(j).tick_id,'1','B','blue'),
                     NULL,
                     'ModTT'||TO_CHAR(lar_tt_rpt(j).tick_id)||'"')
                );
            ELSIF (ias_col_nm(i) = 'ugrp_cd') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).ugrp_cd,'1') );
            ELSIF (ias_col_nm(i) = 'assoc_cd') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).assoc_cd,'1') );
            ELSIF (ias_col_nm(i) = 'status_nm') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).status_nm,'1') );
            ELSIF (ias_col_nm(i) = 'call_type_nm') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).call_type_nm,'1') );
            ELSIF (ias_col_nm(i) = 'prob_type_nm') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).prob_type_nm,'1') );
            ELSIF (ias_col_nm(i) = 'author_nm') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).author_nm,'1') );
            ELSIF (ias_col_nm(i) = 'tick_type') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).tick_type,'1') );
            ELSIF (ias_col_nm(i) = 'svc_type_nm') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).svc_type_nm,'1') );
            ELSIF (ias_col_nm(i) = 'priority_cd') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).priority_cd,'1') );
            ELSIF (ias_col_nm(i) = 'open_dtm') THEN
               ia_ui.tcell_str( ia_ui.s(TO_CHAR(lar_tt_rpt(j).open_dtm,c.DATETIMEMASK),'1') );
            ELSIF (ias_col_nm(i) = 'close_dtm') THEN
               ia_ui.tcell_str( ia_ui.s(TO_CHAR(lar_tt_rpt(j).close_dtm,c.DATETIMEMASK),'1') );
            ELSIF (ias_col_nm(i) = 'closer_nm') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).closer_nm,'1') );
            ELSIF (ias_col_nm(i) = 'tick_num') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).tick_num,'1') );
            ELSIF (ias_col_nm(i) = 'country_nm') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).country_nm,'1') );
            ELSIF (ias_col_nm(i) = 'sub_type') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).sub_type,'1') );
            ELSIF (ias_col_nm(i) = 'trg_cd') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).trg_cd,'1') );
            ELSIF (ias_col_nm(i) = 'skip_flg') THEN
               ia_ui.tcell_str( ia_ui.s(lar_tt_rpt(j).skip_flg,'1') );
            ELSIF (ias_col_nm(i) = 'cycle_tm') THEN
               ia_ui.tcell_str( ia_ui.s(utils.fs_ifnull(lar_tt_rpt(j).cycle_tm,'', 
                     TO_CHAR(lar_tt_rpt(j).cycle_tm)||' hrs'),'1') );
            ELSIF (ias_col_nm(i) = 'age') THEN
               IF (lar_tt_rpt(j).status_nm = 'CLOSED') THEN
                  ia_ui.tcell_str('N/A');
               ELSE
                  ia_ui.tcell_str( ia_ui.s(utils.fs_ifnull(lar_tt_rpt(j).age,'', 
                     TO_CHAR(lar_tt_rpt(j).age)||' hrs'),'1') );
               END IF;
            ELSIF (ias_col_nm(i) = 'contact_age') THEN
               IF (lar_tt_rpt(j).status_nm = 'CLOSED') THEN
                  ia_ui.tcell_str('N/A');
               ELSE
                  ia_ui.tcell_str( ia_ui.s(utils.fs_ifnull(lar_tt_rpt(j).contact_age,'', 
                     TO_CHAR(lar_tt_rpt(j).contact_age)||' hrs'),'1') );
               END IF;
            ELSE
               ia_ui.tcell_str( ia_ui.s('Unknown Column','1'));   
            END IF;
         END LOOP;
         
         htp.tableRowClose;
         
      END LOOP;

   htp.tableClose;
   htp.formClose;
      
   ia_ui.report_footer;
   ia_ui.footer;

END tt_search_rslt;

